Procedimentos Graficos em Calculo Integral

Universidade Estadual de Maringá
Departamento de Matemática
Prof. Doherty Andrade (DMA- UEM)
Prof. Timothy M. (WLU-USA)

Maple

Este procedimento plota a soma de Riemann em 3 dimensões. Você pode conjugar o gráfico da função com a soma de Riemann em 3 dimensoes.

Execute o procedimento e faça os exemplos.

O Procedimento (execute-o)

Exemplos

```
> riemann3d(u^2+v^2,u=0..1,v=0..1);
```



```
> riemann3d(exp(-u^2-v^2),u=-1..1,v=0..1,grid=[20,10]);
```



> **riemann3d(sin(x)*sin(y),x=0..Pi,y=0..Pi,grid=[20,10]);**



> **riemann3d(x*y,x=0..4,y=0..4,grid=[20,10]);**



No próximo exemplo vamos plotar uma superficie e sob ela os paralelepipedos que aproximam o seu volume.

```
> with(plots):

> p1:=plot3d(sin(x)*sin(y),x=0..Pi,y=0..Pi):

> p2:=riemann3d(sin(x)*sin(y),x=0..Pi,y=0..Pi,grid=[20,10]):

> plots[display3d]({p1,p2});
```



Outro exemplo uma superficie e sob ela os paralelepipedos que aproximam o seu volume.

```
> q1:=riemann3d(x*y,x=0..4,y=0..4,grid=[20,10]):
> q2:=plot3d(x*y,x=0..Pi,y=0..Pi):
> plots[display3d]({q1,q2});
```



>

O Procedimento (execute-o)

```
> riemann3d := proc ()
> local f,r1,r2,a,b,c,d,v1,v2,n,m,i,j,L,step1,step2,v;
> r1 := args[2];
> r2 := args[3];
> a := convert(convert(r1,list)[2],list)[1];
> b := convert(convert(r1,list)[2],list)[2];
> c := convert(convert(r2,list)[2],list)[1];
> d := convert(convert(r2,list)[2],list)[2];
> v1 := convert(r1,list)[1];
> v2 := convert(r2,list)[1];
> f := unapply(args[1],v1,v2);
> # default grid
> n := 10;
> m := 10;
> for i from 3 to nargs
> do
> if type(args[i],`=`) then
> if convert(args[i],list)[1]=grid then
> n := convert(args[i],list)[2][1];
> m := convert(args[i],list)[2][2];
> fi;
> fi;
> od;
> step1 := (b-a)/n;
> step2 := (d-c)/m;
```

```
> L := plot3d([0,0,0],r1,r2):
> for j from 0 to m-1 do
> for i from 0 to n-1 do
> v := f(a+i*step1+step1/2,c+j*step2+step2/2);
> L := L,plots[polygonplot3d]([[a+step1*i,c+step2*j,0],
> [a+step1+step1*i,c+step2*j,0],[a+step1+step1*i,c+step2*j,v],
> [a+step1*i,c+step2*j,v]],style=patch):
> L := L,plots[polygonplot3d]([[a+step1*i,c+step2*j,0],
> [a+step1*i,c+step2+step2*j,0],[a+step1*i,c+step2+step2*j,v],
> [a+step1*i,c+step2*j,v]],style=patch):
> L := L,plots[polygonplot3d]([[a+step1+step1*i,c+step2*j,0],
> [a+step1+step1*i,c+step2+step2*j,0],
> [a+step1+step1*i,c+step2+step2*j,v],
> [a+step1+step1*i,c+step2*j,v]],style=patch):
> L := L,plots[polygonplot3d]([[a+step1+step1*i,c+step2+step2*j,0],
> [a+step1*i,c+step2+step2*j,0],[a+step1*i,c+step2+step2*j,v],
> [a+step1+step1*i,c+step2+step2*j,v]],style=patch):
> L := L,plots[polygonplot3d]([[a+step1*i,c+step2*j,v],
> [a+step1+step1*i,c+step2*j,v],[a+step1+step1*i,c+step2+step2*j,v],
> [a+step1*i,c+step2+step2*j,v]],style=patch):
> od;
> od;
> plots[display3d]([L]);
> end:
```